    MOODULID, MOODULPROGRAMMEERIMINE�PRIVATE ��


               Tehnika "suureks programmeerimiseks"  





MOODUL:  





== suure programmi osadekaupa töötlemine





ÚÄÄÄÄÄÄÄÄÄÄÄ¿       ÚÄÄÄÄÄÄÄÄÄÄÄ¿  ÚÄÄÄÄÄÄÄÄÄÄÄ¿  ÚÄÄÄÄÄÄÄÄÄÄÄ¿


³ PROGRAM P ³       ³ PROGRAM P ³  ³ Moodul M1 ³  ³ Moodul M2 ³


³           ³       ³ .         ³  ³ .         ³  ³ .         ³


³           ³       ³ .         ³  ³ .         ³  ³ .         ³


³           ³       ³ .         ³  ³ .         ³  ³ .         ³


³           ³       ³ END.      ³  ³           ³  ³           ³


³           ³       ÀÄÄÄÄÄÄÄÄÄÄÄÙ  ÀÄÄÄÄÄÄÄÄÄÄÄÙ  ÀÄÄÄÄÄÄÄÄÄÄÄÙ


³           ³                                   


³           ³       P, M1, M2 töödeldakse eraldi: 


³           ³         -- sisestatakse eraldi


³           ³         -- redigeeritakse eraldi


³           ³         -- transleeritakse eraldi


³           ³         -- salvestatakse eraldi


³ END.      ³         -- testitakse eraldi


ÀÄÄÄÄÄÄÄÄÄÄÄÙ         -- saavad programmi mälukaardil 


                         eraldi mälupiirkonnad (<= 64 KB)








== temaatiline teenustekogum.  





   Näit: -- moodul tööks kompleksarvudega  


         -- graafikaoperatsioonide (süsteemne) moodul Graph


         -- moodul magasinioperatsioonideks 


         -- ...   ...   ...





== programmeerimise stiil -- moodulprogrammeerimine  





   toetab tööd "jaga ja valitse" pôhimôttel: 


   igal momendil tegeleme "väikeste (lihtsate) asjadega"











MOODULI STRUKTUUR





Loogiliselt:              Moodul


                            ³


      ÚÄÄÄÄÄÄÄÄÄÄÄÄÄÄÄÄÄÄÄÄÄÅÄÄÄÄÄÄÄÄÄÄÄÄÄÄÄÄÄÄÄÄÄ¿


      ³                     ³                     ³


Teenuste osa       Realisatsiooniosa      Algväärtustamisosa


ehk interfeis      Teenuste               Mooduli muutujatele


Teenuste kirjel-   realisatsioon          algväärtuste


dused (m.h. AP-de  (AP-de täielikud       omistamine


päised)            tekstid jm.)








Füüsiliselt vôib realisatsiooniosa olla (koos algväärtustamisosa�ga) tekstuaalselt omaette olev eraldi töödeldav programmiük�sus. 





Teenused:�PRIVATE ��





-- konstandid


-- tüübid


-- muutujad


-- alamprogrammid





Skoobireegid:





-- interfeisis teenusteks olevad identifikaatorid kehtivad ka real-osas ja algväärtustamisosas, vastupidine ei kehti: realisatsiooniosas deklareeritud ident-id ei kehti interfeisis (küll aga kehtivad nad algväärtustamisosas)


-- USES-fraasiga interfeisi käsutusse tellitud ident-id ei kehti realisatsiooniosas


-- kui mingit moodulit kasutavas programmiüksuses (näiteks pôhiprogrammis) esinev ident. ühtib kasutatavas moodulis olevaga, kehtibselles üksuses temas endas deklareeritud ident. Soovi korral kasutada kasutatava mooduli ident-it tuleb kasutada liitnime: moodulinimi.ident





Näide. 





PROGRAM Naide;                 UNIT Mood;


USES Mood;   <ÄÄÄÄÄÄÄ¿         INTERFACE


CONST N = 10;        ³      ³  CONST N = 1000;


TYPE                 ³      ³  TYPE


  T = ...;           ÀÄÄÄÄÄÄ´    T = ARRAY[1..N]OF Byte;


  S = ...;        Teenused  ³  PROCEDURE P(A:T; ...): Boolean; 


VAR                         ³  FUNCTION F(...);


  A,B: ...;                    IMPLEMENTATION 


  Q: T;                     ³  PROCEDURE Abi (...);


BEGIN                       ³  ...


  ...                       ³  END; 


  M := N;                   ³ PROCEDURE P;


  P(...);                   ³ VAR I: ...;


  IF	Mood.F(...) THEN       ³ BEGIN


  ...                       ³   ...


  Writeln(Mood.N);          ³ END;                       


  ...                       ³  FUNCTION F...: Boolean;


END.                        ³  ...


                            ³  END;


                              


                               END. {Mood}








Mooduli erijuht -- identifikaatorideklaratsioonide moodul:





UNIT nimi;


INTERFACE


  Identifikaatorideklaratsioonid





IMPLEMENTATTION


END.














KEELEREEGLITE TÄPSUSTUSI�PRIVATE ��








Mooduli üldkuju:





UNIT moodulinimi;


INTERFACE                            Liides on mooduli ligipää-


  [USES mood, mood, ...;] – kasut.   setav osa (kättesaadav osa


                          liideses   avalik osa, nähtav osa


 Konstandi-, tüübi- ja              (public part, visible part)


 muutujadekl-d ning 


 AP-päised    





IMPLEMENTATION                       Real.osa on mooduli peide-


  [USES mood, mood, ...;] -- kasut.  tud osa -- hidden part,


                          real.osas  (salastatud osa, eraosa - 


  [Dekl-d,                            private part, mittekätte-


  sh. märgendidekld                   saadav osa)


      tüübidekl-d


      muutujadekl-d


      konstandidekl-d


      AP dekl-d]


  [BEGIN                              Algv.osa: (liidese vôi  


    Laused]                           realosa) muutujate väär-


END.                                  tustamine toimub ainult 


                                      üks kord - programmi


                                      täitmise algul








Tarbijale (ka moodulit kasutavale programmiüksusele) on     kättesaadav ainult mooduli teenusteosa, real-osale ta ligi ei  pääse.














Moodulprogrammi (mooduleid sisaldava programmi) töötlemine





-- moodul sisestatakse, salvestatakse, transleeritakse eraldi


-- mooduli nimi peab ühtima vastava faili nimega!


-- transleeritud moodulifail on TPU-fail


-- enne tuleb transleerida moodul, siis teda kasutav programm


-- programmile saab rakendada Make'i -- teeb kôik vajalikud tööd


-- programmi (edukal) transleerimisel tekib EXE-fail


-- moodulite kaudsel (teise mooduli kaudu) kasutamisel tuleb näidata vähemalt otseselt kasutatava mooduli nimi:





PROGRAM PR;        UNIT M1;          UNIT M2;


USES M1;           INTERFACE         INTERFACE


   {vôi M1, M2}    USES M2;











       MOODULID ERI KEELTES�PRIVATE ��





 Turbo Pascal           Modula-2                Ada


    Unit                 Module               Package





UNIT nimi;            DEFINITION MODULE..;  PACKAGE nimi IS


INTERFACE             EXPORT ...;           Teenused


Teenused              Teenused ehk           ** (Abi)dekl-d


  ** Dekl-nid           eksportartiklid      ** AP-de dekl-d


  ** AP-de päised       ** dekl-d,            [ PRIVATE


.                       **  AP-de päised        dekl-d ]


.                      . . .                . . .


.                     END nimi;             END nimi;


- - - - - - -      =====================   =================


{ Real.osa: }     { Real.moodul: }         -- Paketi sisu:


IMPLEMENTATION    IMPLEMENTATION MODULE..; PACKAGE BODY ...IS


** (Abi)dekl-nid     ** (Abi)dekl-d         ** (Abi)dekl-d 


** AP-de dekl-nid    ** AP-de dekl-nid      ** AP-de dekl-d


   (täistekstid)                            


- - - - - - - - -      - - - - - - - -      - - - - - - - -  


[ BEGIN  { Algv.osa}   [ BEGIN              [ BEGIN


 Algväärt.laused ]      Algv.laused ]        Algv-laused ]  


                                            - - - - - - - - 


                                            [ EXCEPTION


                                              Eriolukordade


                                                töötlus ]


END.                   END;                  END nimi;


===========================================================


PROGRAM nimi;          MODULE nimi;          USE paketi-nimi;


USES mood-nimi;       FROM nimi IMPORT...;   PROCEDURE nimi; 


.                       (mooduli teenuste    . 


.                        import)             .


.                      . . .                 . 


END.                   END nimi;             END nimi; 


===========================================================


NB!   Turbo Pascal  <  Modula-2  <<  Ada





                   STANDARDMOODULID	�PRIVATE ��





Need on valmismoodulid,  mille transleeritud tekstid on failis Turbo.TPL. See fail kantakse Turbo Pascalisse minekul automaatselt pôhimällu ja ta paikneb seal Turbo Pascalist väljumiseni.





Stand.moodulid sisaldavad sageli vajatavaid temaatilisi lisateenuseid - nad laiendavad keelt. Standardmooduli kasutuselevôtuks on, nagu ikka, USES-fraas.





Stand.moodulite kogum on versiooniti erinev ja on süsteemi kasutaja poolt lihtsalt muudetav - abiprogrammi TPUMover abil.


Alati on olemas (nad kôik ei pruugi alati Turbo.TPL-is olla): 





System - täitmisaja teek


Dos - opsüsteemi esindaja T. Pascalis


Printer - väljastuseks printeril


Crt - ekraani ja sôrmistiku juhtimiseks


Graph - graafika


Overlay - ülekattestruktuurid


Turbo3 - seoseks vana versiooniga 3.0


Graph3 - kilpkonnagraafika (Turtle graphics)





System  - täitmisaja teek (run-time library): on seostatud iga programmiga (USES't pole vaja) ja mooduliga.


System sisaldab:


- standardalamprogrammid (nende koodid)


- madaltasemevahendid


- S/V-vahendid


- ujupunktitehnika


- . . . 


- hulganisti muutujaid (sh. 23 tüpiseeritud konstanti ja


                            20 "tavalist" muutujat)








Näiteid:


CONST


  OvrHeapSize: Pointer = NIL;  { Ülekattepuhvri algmôôt }


  ErrorAddr: Pointer = NIL;    { Hiipiviit }


  PrefixSeg: Word = 0;         { Täitmisaja vea aadress }


  . . .                         


VAR


  Input: Text;                 { Standardne sisendfail }


  Output: Text;                { Standardne väljundfail }





Dos  - on opsüsteemi "esindaja" Turbo Pascalis: sisaldab opsüsteemi käskude analooge ning konstante, tüüpe, muutujaid.


Näiteid:


TYPE


  Time = RECORD


   Year, Month, Day, Hour, Min, Sec: Integer;


  END;


VAR


  DosError: Integer;     { = 0 - viga pole, 


                           = 2 - faili ei leitud, . . . }


. . .


PROCEDURE GetDate (VAR Year, Month, Day, DayOfWeek: Word);


   "      GetTime (VAR Hour, Minute, Second, Sec100: Word);


FUNCTION DiskFree(Drive: Word): Longint;


   "     DiskSize(Drive: Word): Longint;





Printer - väljastuseks printeril. Sisaldab tektifaili Lst deklaratsiooni, seotuse printeriga Lpt1, faili avamise ja sulgemise operatsioonid.


PROGRAM . . . ;


USES  Printer;


. . .


Writeln(Lst, väljundinfo);


. . .








TPUMover : c:\. . .\TP\TPUMOVER[Turbo.TPL[+-TPU-fail]]


                             (+ lisamine   - eemaldamine)























{*****************************************************}�PRIVATE ��


{*                                                   *}


{*     MOODUL RatsArv osutab järgmisi teenuseid      *}


{*                ratsionaalarvudega:                *}


{*       � ratsionaalarvude moodustamine             *}


{*       �      �� " ��     sisestamine              *}


{*       �      �� " ��     väljastamine             *}


{*       �      �� " ��     korrutamine              *}


{*       �      �� " ��     jagamine                 *}


{*       �      �� " ��     liitmine                 *}


{*       �      �� " ��     lahutamine               *}


{*       �      �� " ��     vôrdsuse kontroll        *}


{*       �      �� " ��     kontroll "väiksem"       *}


{*                                                   *}


{*****************************************************}


UNIT RatsArv;





INTERFACE                                 { INTERFACE }





CONST  MaxInt = 32767;





TYPE


  Positiiv = 1..MaxInt;


  Rats = RECORD


    Lug: Integer;


    Nim: Positiiv;


  END;





PROCEDURE MoodustaRats(Tais: Integer; Pos: Positiiv;


                       VAR X: Rats);


PROCEDURE SisestaRats(VAR X: Rats);


PROCEDURE ValjastaRats(X: Rats);


PROCEDURE Korruta(X, Y: Rats; VAR Kor: Rats);


PROCEDURE Jaga(X, Y: Rats; VAR Jag: Rats);


PROCEDURE Liida(X, Y: Rats; VAR Sum: Rats);


PROCEDURE Lahuta(X, Y: Rats; VAR Vahe: Rats);


FUNCTION Vordub(X, Y: Rats): Boolean;


FUNCTION Vaiksem(X, Y: Rats): Boolean;





IMPLEMENTATION                       { IMPLEMENTATION }


                                 { Abideklaratsioonid }


TYPE


  YksKuniViis = 1..5;





FUNCTION Pikkus(Arv: Positiiv): YksKuniViis;


BEGIN


  CASE Arv OF


  0..9:           Pikkus := 1;


  10..99:         Pikkus := 2;


  100..999:       Pikkus := 3;


  1000..9999:     Pikkus := 4;


  10000..MaxInt:  Pikkus := 5


  END;


END;  {Pikkus}











PROCEDURE YhisNimetaja(VAR X, Y: Rats);


VAR


  YhNim: Positiiv;


BEGIN


  YhNim := X.Nim*Y.Nim;


  X.Lug := X.Lug*Y.Nim;


  Y.Lug := Y.Lug*X.Nim;


  X.Nim := YhNim;


  Y.Nim := YhNim;


END;  {YhisNimetaja}





FUNCTION SuurimYhistegur(Arv1, Arv2: Positiiv): Positiiv;


VAR


  A1, A2: Positiiv;


BEGIN


  A1 := Arv1;


  A2 := Arv2;


  WHILE A1 <> A2 DO


    IF A1 > A2 THEN


      Dec(A1, A2)


    ELSE


      Dec(A2, A1);


  SuurimYhistegur := A1;


END;  {SuurimYhistegur}





PROCEDURE Taanda(VAR X: Rats);


VAR


  SYT: Word;


BEGIN


  SYT := SuurimYhistegur(Abs(X.Lug), X.Nim);


  X.Lug := X.Lug DIV SYT;


  X.NIm := X.Nim DIV SYT;


END;  {Taanda}





           { Teenuste�alamprogrammide deklaratsioonid }


PROCEDURE MoodustaRats;


BEGIN


  X.Lug := Tais;


  X.Nim := Pos;


END;  {MoodustaRats}





PROCEDURE SisestaRats;


BEGIN


  Write('Sisesta lugeja (koos märgiga): ');


  Readln(X.Lug);


  Write('Sisesta nimetaja (positiivne): ');


  Readln(X.Nim);


  WHILE X.Nim <= 0 DO


  BEGIN


    Write('Nimetaja pole positiivne. Sisesta uuesti: ');


    Readln(X.Nim);


  END;


Write('Lôpus anna ^z ja Enter; kui arve on veel,siis Enter: ');


END;  {SisestaRats}











PROCEDURE ValjastaRats;





VAR


  LugPikkus, NimPikkus: YksKuniViis;





BEGIN





  LugPikkus := Pikkus(Abs(X.Lug)) + 2;


  NimPikkus := Pikkus(X.Nim) + 1;


  Write(X.Lug:LugPikkus, ' /', X.Nim:NimPikkus);





END;  {ValjastaRats}





PROCEDURE Korruta;


BEGIN





  Kor.Lug := X.Lug*Y.Lug;


  Kor.Nim := X.Nim*Y.Nim;


  Taanda(Kor);





END;  {Korruta}





PROCEDURE Jaga;


BEGIN





  Jag.Lug := X.Lug*Y.Nim;


  Jag.Nim := X.Nim*Y.Lug;


  Taanda(Jag);





END;  {Jaga}




















PROCEDURE Liida;





VAR


  X1, Y1: Rats;





BEGIN





  X1 := X;


  Y1 := Y;


  YhisNimetaja(X1, Y1);


  Sum.Lug := X1.Lug + Y1.Lug;


  Sum.Nim := X1.Nim;


  Taanda(Sum);





END;  {Liida}




















PROCEDURE Lahuta;





VAR


  X1, Y1: Rats;





BEGIN





  X1 := X;


  Y1 := Y;


  Y1.Lug := Y1.Lug*(�1);


  Liida(X1, Y1, Vahe);





END;  {Lahuta}





FUNCTION Vordub;





VAR


  X1, Y1: Rats;





BEGIN





  X1 := X;


  Y1 := Y;


  YhisNimetaja(X1, Y1);


  IF X1.Lug = Y1.Lug THEN


    Vordub := TRUE


  ELSE


    Vordub := FALSE;





END;  {Vordub}








FUNCTION Vaiksem;





VAR


  X1, Y1: Rats;





BEGIN





  X1 := X;


  Y1 := Y;


  YhisNimetaja(X1, Y1);


  IF X1.Lug < Y1.Lug THEN


    Vaiksem := TRUE


  ELSE


    Vaiksem := FALSE;





END;  {Vaiksem}





END.  { RatsArv }                      { Mooduli lôpp }























{*************************************************************}�PRIVATE ��


{*                                                           *}


{*  PROGRAMM  sisestab tundmatu hulga ratsionaalarve         *}


{*      (kahe täisarvu paarid, millest teine on positiivne)  *}


{*      ning:                                                *}


{*    � liidab kôik arvust 1/2 erinevad arvud;               *}


{*    � korrutab kôik sisestatud arvud                       *}


{*    � leiab 1/2�ga vôrduvate rats.arvude arvu              *}


{*                                                           *}


{*************************************************************}





PROGRAM RatsKas;





USES


  RatsArv;





VAR


  PoolteArv: Word;


  R, Pool, Null, Yks, Summa, AbiSum, Korrutis: Rats;





BEGIN





  MoodustaRats(1, 2, Pool);     { Rats.arv Pool on 1/2 }


  MoodustaRats(1, 1, Yks);      { Rats.arv Yks on 1/1 }


  MoodustaRats(0, 1, Null);     { Rats.arv Null on 0/1 }


  PoolteArv := 0;


  Summa := Null;                     { Kirjete omistus }


  Korrutis := Yks;





  Writeln;


  Write('Alustame tööd ratsionaalarvudega, vajuta Enter�ile');





  WHILE NOT Eof DO     (* Faili lôpuni *)


  BEGIN


    SisestaRats(R);


    Korruta(R, Korrutis, Korrutis);


    IF Vordub(R, Pool) THEN


      Inc(PoolteArv)


    ELSE


      BEGIN


        Liida(Summa, R, AbiSum);


        Summa := AbiSum


      END;


  END;





  Writeln;


  Write('Korrutis on järgmine: ');


  ValjastaRats(Korrutis);


  Writeln;


  Writeln('1/2�ga vôrdseid rats.arve on: ', PoolteArv:3);


  Write('Summaks on rats.arv  ');


  ValjastaRats(Summa);


  Writeln;


END.    (* RatsKas *)











{**********************************************************}�PRIVATE ��


{*                                                        *}


{*   MOODUL  ArvuHoid  töötab FiFo pôhimôttel,            *}


{*           hoidlasse saab tarve lisada ja eemaldada     *}


{*           (protseduurid Lisa ja Eemalda).              *}


{*           Muutuja Tyhi = TRUE, kui hoidlas arve pole,  *}


{*           ja MitteTais = TRUE, kui hoidlas on ruumi.   *}


{**********************************************************}





UNIT ArvuHoid;





INTERFACE                                    {* INTERFACE *}


VAR


  Tyhi, MitteTais: Boolean;


PROCEDURE Lisa(Arv: Integer);


PROCEDURE Eemalda(VAR Arv: Integer);





IMPLEMENTATION                          {* IMPLEMENTATION *}


CONST


  N = 200;


  M = 199;


VAR


  Esimene, Viimane: 0..M;


  MituOn :        0..N;


  Mas: ARRAY[0..M] OF Integer;





PROCEDURE Lisa;


BEGIN


  IF MituOn < N THEN


  BEGIN


    Mas[Viimane] := Arv;


    Viimane := (Viimane + 1) MOD N;


    Inc(MituOn);


    MitteTais := MituOn < N;


    Tyhi := FALSE


  END


END;   { Lisa }





PROCEDURE Eemalda;


BEGIN


  IF MituOn>0 THEN


  BEGIN


    Arv := Mas[Esimene];


    Esimene := (Esimene + 1) MOD N;


    Dec(MituOn);


    MitteTais := TRUE;


    Tyhi := MituOn = 0


  END


END;   { Eemalda }


                                      {* Algväärtustamine *}


BEGIN


  MituOn := 0;


  Esimene := 0;


  Viimane := 0;


  Tyhi := TRUE;


  MitteTais := TRUE


END.  { ArvuHoidla }                      {* Mooduli lôpp *}


{************************************************************}�PRIVATE ��


{*                                                          *}


{*    PRORAMM kannab arvuhoidlasse tundmatu hulga täisarve  *}


{*            ning eemaldab ja väljastab need.              *}


{*                                                          *}


{*            Programm kasutab moodulit ArvuHoid            *}


{*                                                          *}


{************************************************************}





PROGRAM MangArvuHoidlaga;





USES


  ArvuHoid;





  VAR


    Arv: Integer;


    Loendaja: Word;





BEGIN





                               {* Arvude kandmine hoidlasse *}


  Writeln;


  Write('Kui oled sisestuseks valmis, vajuta Enter. ');





  WHILE NOT Eof DO


  BEGIN





    Write('Sisesta arv : ');


    Readln(Arv);


    Write('Lôpus ^z + Enter, kui arve on veel, siis Enter.');





    IF MitteTais THEN


      Lisa(Arv)


    ELSE


      BEGIN


        Writeln;


        Write('Hoidla on täis, lôpeta töö (^z + Enter). ');


      END;





  END;  {WHILE}





                                     {* Hoidla tühjendamine *}


  Loendaja := 0;





  WHILE NOT Tyhi DO


  BEGIN


    Eemalda(Arv);


    Inc(Loendaja);


    Writeln('Eemaldatud arv on : ', Arv:5)


  END;





  Writeln('Hoidlas oli ', Loendaja:4, ' arvu. ');


  Writeln;


END.   {* MangArvuHoidlaga *}











Ülesanne moodulite koostamise kohta: Koostada moodul, mille teenusteks on vähemalt järgmised operatsioonid:                �PRIVATE ��


                                                + TESTPROGRAMM!


(1) Operatsioonid kolmnurkadega:


-- kas on kolmnurk


-- pindala


-- ümbermôôt


-- mis liiki (nürinurkne, teravnurkne, ...)


-- ümberjoonestatud ringjoone pindala


-- sissejoonestatud ringjoone pindala


-- ...


(2) Hulkade S/V-operatsioonid täisarvuhulkade (SET OF Byte) ja märgihulkade (SET OF Char) jaoks.


(3) Maatriksioperatsioonid  +  -  *  S/V


(4) Dialoogsisestuse ja väljastuse protseduurid vektorite jaoks.


(5) Dialoogsisestuse ja väljastuse prots-rid maatriksite jaoks.


(6) Kuupäevaoperatsioonid:


-- dialoogsisestus                       RECORD


-- väljastus                               paev: ...


-- päevade arv teatud kuupäevani           kuu: ...


-- päevade arv aasta algusest              aasta: ...


-- ...                                   END


(7) Operatsioonid kompleksarvudega:


-- liitmine                              RECORD


-- lahutamine                              ReX: ...


-- korrutamine                             ImX: ...


-- sisestus ja väljastus                 END


-- ruut


-- ...


(8) Vektoritehted:    +  -  *  vektorkorrutis  S/V


(9) Operatsioonid täisarvupaaridega:


-- sisestus                              RECORD


-- väljastus                               a: Integer


-- suurim ühistegur                        b: Word


-- vähim ühiskordne                      END


-- taandamine (SÜT-iga jagamine)


-- ...


(10) Magasinioperatsioonid   Push  Pop  Empty  ...


(11) Operatsioonid temperatuurikogumikega


     Temp[Jaan..Dets, 1..31, 0..23] OF Integer :


-- maksimaaltemp. (teatud kuul, kokku, ...)


-- keskmine temp. (teatud kuul, ...)


-- teatud tabeliosade väljastus


-- ...                 >= 7 operatsiooni !


(12) Operatsioon(id) lineaarvôrrandisüsteemi lahendamiseks.


(13) Operatsioon(id) pöördmaatriksi leidmiseks.


(14) Standard-Pascalis esineva sisefailitehnika (S/V-operatsioo�nide) ühemôôtmelisel massiivil imiteerimise operatsioonid.


(15) Operatsioonid ruutvôrrandi ax2 + bx + c = 0 lahendamiseks -- kôik juhud: reaallahendid, komplekslahendid, kôdunud juhud a,b ja/vôi c = 0. Moodul sisaldagu ka suhtlemise tarbijaga.


(16) Operatsioonid kuupvôrrandi ax3+bx2+cx+d = 0 lahendami�seks.


(17) Lihtsamate geomeetriliste kujundite joonistamise oper-nid (suvaliste programmiliste vahenditega, va. Graph'i prots-d).

